BOX ... COORDINATES

PURPOSE
Specifies the coordinates of a box to appear on subsequent plots.

DESCRIPTION
The 2 pairs of coordinates define the (x,y) values for opposing corners of the box. The most common uses of this command are to draw boxes around legends and in creating legends for bar charts.

SYNTAX
BOX <id> COORDINATES <x1> <y1> <x2> <y2>
where <id> is an integer number or parameter in the range 1 to 100 that specifies the box identifier;
<x1> is a decimal number or parameter in the decimal range 0 to 100 that specifies the x coordinate of one box corner;
<y1> is a decimal number or parameter in the decimal range 0 to 100 that specifies the y coordinate of one box corner;
<x2> is a decimal number or parameter in the decimal range 0 to 100 that specifies the x coordinate of the other box corner;
and <y2> is a decimal number or parameter in the decimal range 0 to 100 that specifies the y coordinate of the other box corner.

EXAMPLES
BOX 1 COORDINATES 50 50 60 60
BOX 2 COORDINATES 20 60 40 80

NOTE 1
Boxes can be generated in two ways. You can use the BOX COORDINATES command to place boxes on subsequent plots or you can use the BOX command to draw one box immediately. In either case, the attributes of the box border are set with the BOX COLOR, BOX PATTERN and the BOX THICKNESS commands while the attributes of the box interior are set with the BOX FILL COLOR, BOX FILL PATTERN, BOX FILL GAP, BOX FILL LINE, and BOX FILL THICKNESS commands.

You can also use the CHARACTER command to set specific traces to a box (e.g., CHARACTER BOX CIRCLE) and use the various CHARACTER attribute setting commands to set the attributes for the box. The distinction between this and the BOX COORDINATES command is that you can use plot coordinates rather than DATAPLOT screen coordinates. The following shows how you might do something like this (assume Y1, Y2, and X contain your data):

LINE OFF ALL
BAR ON ON
BAR COLOR BLACK G50
CHARACTER OFF OFF BOX BOX
CHARACTER HW 4 4 ALL
CHARACTER FILL ON ALL
CHARACTER COLOR BLACK BLACK BLACK G50
LET XBOX = DATA 300 300
LET YBOX = DATA 450 425
LET BOXTAG = DATA 3 4
PLOT Y1 Y2 VS X AND
PLOT YBOX XBOX BOXTAG

NOTE 2
A BOX ... COORDINATES command with no arguments omits the box from subsequent plots. Thus BOX 1 COORDINATES with no arguments omits box 1 from subsequent plots. A BOX ... COORDINATES command with no <id> refers to all 100 boxes. Thus BOX COORDINATES 30 50 assigns the coordinates (30,50) to all 100 boxes (but this has no practical use). BOX COORDINATES with no <id> and no arguments omits all 100 boxes from subsequent plots.

DEFAULT
No boxes are drawn.

SYNONYMS
BOX CORNER COORDINATES is a synonym for BOX COORDINATES.
RELATED COMMANDS

- **PLOT** = Generates a data or function plot.
- **BOX** = Draws a box (immediately).
- **BOX COLOR** = Sets the colors for box borders.
- **BOX PATTERN** = Sets the line styles for box borders.
- **BOX THICKNESS** = Sets the line thicknesses for box borders.
- **BOX FILL COLOR** = Sets the colors for box fill patterns.
- **BOX FILL GAP** = Sets the spacings for box hatch patterns.
- **BOX FILL LINE** = Sets the line types for box hatch patterns.
- **BOX FILL PATTERN** = Sets the fill patterns for boxes.
- **BOX FILL THICKNESS** = Sets the line thicknesses for box hatch patterns.
- **LEGEND COORDINATES** = Sets the location of plot legends.
- **ARROW COORDINATES** = Sets the location of plot arrows.
- **SEGMENT COORDINATES** = Sets the location of plot line segments.
- **FRAME COORDINATES** = Sets the location of the plot frame.

APPLICATIONS

Presentation Graphics

IMPLEMENTATION DATE

Pre-1987
PROGRAM
LET CARTER = DATA 66 30 11 43 44 41 35 82 54 36
LET REAGAN = DATA 26 54 84 47 51 51 52 14 36 55
LET X = DATA 1 2 3 5 6 7 8 10 11 12
LET X1 = X - 0.2; LET X2 = X + 0.2
.
YLIMITS 1 12; YTIC OFFSET 1 1; MINOR YTIC MARK NUMBER 0
YTIC LABEL CONTENT DEMOCRATS INDEPENDENTS REPUBLICANS SP() EAST ...
SOUTH MIDWEST WEST SP() BLACKS HISPANICS WHITES SP()
X1LABEL PERCENT; XLIMITS 0 100; MAJOR X Tic NUMBER 11; MINOR X Tic NUMBER 1
.
LINE BLANK BLANK
BAR ON ON
BAR WIDTH 0.2 ALL
BAR DIRECTION H ALL
BAR FILL ON ON
BAR COLOR BLACK G50
BOX FILL PATTERN ON
BOX SHADOW HW 0 0
BOX 1 CORNER COORDINATES 70 80 73 83
BOX 1 FILL COLOR BLACK
BOX 2 CORNER COORDINATES 70 85 73 88
BOX 2 FILL COLOR G50
LEGEND JUSTIFICATION LECE
LEGEND 1 CARTER; LEGEND 1 COORDINATES 74 81.5
LEGEND 2 REAGAN; LEGEND 2 COORDINATES 74 86.5
TITLE DEMONSTRATE BOX COORDINATES COMMAND; TITLE SIZE 5
PLOT X1 CARTER AND
PLOT X2 REAGAN

**DEMONSTRATE BOX COORDINATES COMMAND**

![Chart showing percent distribution of various categories including Democrats, Independents, Republicans, Blacks, Hispanics, Whites, East, South, Midwest, and West.](image-url)